
Założenia wstępne:





1) Sprzęt


Urządzenie powinno posiadać panel LCD o rozmiarach 3x20. Obok panelu


powinna znajdować się klawiaturka na mikroprzełącznikach. Klawiatura


posiada 8 klawiszy:


+10 ; -10 ; +1 ; -1 ; menu ; next ; enter ; cancel





Urządzenie komunikuje się z komputerem za pośrednictwem łącza RS-232. Cały panel powinien składać się z jednej płytki drukowanej, na której jest mikrokontroler, przymocowany wyświetlacz LCD oraz sztywno osadzone mikroprzełączniki tworzące klawiaturke. W rogach płytki drukowanej powinny być tuleje dystansowe, umożliwiające przykręcenie całej płytki do obudowy urządzenia - oczywiście po wcześniejszym wywierceniu otworów na LCD i klawisze. Na płytce należy umieścić niewielkich rozmiarów złącza rozmiarów zworek na płytach głównych na zasilanie oraz na RS-232. 





2) Protokół pomiędzy urządzeniem a PC.


Po uruchomieniu urządzenie nie wyświetla nic - oczekuje na komendę





Pierwsza faza - ściągniecie komunikatów tekstowych do urządzenia


Urządzenie będzie miało możliwość wyświetlania kilku komunikatów


tekstowych, których treść ściągnie sobie z PC-ta.





DOWNLOAD_MESSAGES, ilość, suma kontrolna


Kierunek PC(urządzenie


Parametr ilość można pominąć i założyć stała ilość komunikatów(np. 10). Polecenie to jest wysyłane do urządzenia w celu przekazania z PC do urządzenia komunikatów tekstowych, które będą później wyświetlane





Po odebraniu komunikatu urządzenie odpowiada:





MESSAGE_REQUEST, N, suma kontrolna





W tym momencie PC przesyła do urządzenia treść komunikatu o numerze N.





MESSAGE_REPLY, N, treść komunikatu, END_OF_MESSAGE, suma kontrolna





I tak w pętli aż do przesłania wszystkich komunikatów.


 Urządzenie buforuje komunikaty w pamięci Nie wiem jakie SA rozmiary dostępnych pamięci - 


zasugerujcie ile komunikatów się zmieści Jeśli np. tylko 5 to tez nie ma problemu.





Rozmiar komunikatu = ilość znaków na wyświetlaczu LCD.





W tym momencie urządzenie jest gotowe do pracy, ale nie wyświetla


jeszcze nic na panelu.





Druga faza - normalny tryb pracy urządzenia


Po odebraniu od PC polecenia


RUN, suma kontrolna 





Urządzenie rozpoczyna główna pętle pracy.





Na Panelu LCD wyświetlany jest komunikat powitalny. Naciśniecie przycisku menu spowoduje przejście do następnego menu. Pierwsze menu jest specjalne. Ma następująca formę:





Menu 1)





=====================


| enter IP/mask		|


|      V				|


|149.156.100.100/24		|


=====================








Pozostałe menu zawierają komunikaty, które zostały przesłane


z PC poleceniem MESSAGE_REPLY.





Pierwsze menu umożliwia wprowadzanie danych, pozostałe tylko wyświetlają komunikaty. Menu 1 działa następująca Po wejściu do menu pojawia się adres IP w postaci 0  0  0  0 / 0. Nad pierwszym zerem pojawia się ptaszek "V". Wartość elementu wskazywanego przez ptaszek można zmieniać przyciskami +-10 +-1. Następnie możemy przesunąć się do kolejnego segmentu adresu IP przyciskiem net Naciśniecie net na ostatnim segmencie powoduje przejście do pierwszego. Gdy użytkownik wprowadzi już adres IP naciska klawisz eter W tym momencie urządzenie przesyła komunikat do PC:





SET_IP_ADDRESS, ip/mask, suma_kontrolna 





Adres IP/maska mogą być w formie tekstowej czy tez binarnej - jak wygodniej będzie zaprogramować Jeśli nie będzie zbyt trudno sprawdzić poprawność adresu IP to należy to zrobić w mikrokontrolerze. Jeśli będzie to trudne, to urządzenie przesyła IP do PC a ten potwierdza negatywnie lub pozytywnie. 





PC odpowiada 





IP_OK - urządzenie wyświetla komunikat, ze adres został ustawiony i wraca do stanu początkowego (wyświetla komunikat powitalny)





IP_ERROR - urządzenie wyświetla komunikat, ze adres jest niepoprawny


i wraca do menu 1.





Pozostałe menu są już prostsze. Przejście do kolejnego menu powoduje wyświetlenie komunikatu, którego treść została wcześniej przesłana do mikrokontrolera. Przykładowym komunikatem może być "Numer of sensors?". Jest teraz użytkownik naciśnie eter,


to do PC zostanie przesłana informacja:





USER_COMMAND, nr_polecenia, suma_kontrolna





PC odbierze komunikat, wykona pewne operacje i ich wynik przęśle w formie tekstowej do mikrokontrolera. Urządzenie wyświetla komunikat i po pewny czasie wraca do stanu początkowego Numer polecenia odpowiada oczywiście numerowi downloadowanego


we wcześniejszej fazie komunikatu.























Po zapoznaniu się z wymaganiami zmienione zostały następujące elementy specyfikacji:





Z uwagi na dostępność zastosowano wyświetlacz LCD 4x16 znaków oparty na standardowym sterowniku wyświetlaczy LCD HD44780. Sterownik ten tworzy zamknięty moduł wyświetlacza komunikujący się z systemem nadrzędnym poprzez 7 lub 11 linii.


Uniwersalność modułu wyświetlacza pozwala go zastąpić modułem o innej organizacji pola znakowego.


Zmieniono definicje funkcji klawiszy na następującą:


	


Menu�
Help�
Cancel�
Enter�
�
(�
(�
(�
(�
�



Wprowadzono możliwość sprowadzania konsoli do stanu początkowego przez komputer wykorzystując do tego linie RTS (CTS po stronie konsoli).


Przygotowano opcjonalnie miejsce pod szeregową pamięć I2C


Zdefiniowano protokół wymiany komunikatów spełniający zakładane wymagania.








�
Opis konstrukcji





Sercem konsoli jest mikrokontroler jednoukładowy AT89c52. Posiada on 8kB wewnętrznej pamięci programu typu Flash , 256B wewnętrznej pamięci danych, wbudowane układy czasowo-licznikowe oraz interfejs szeregowy. Programowo jest on zgodny z mikrokontrolerem 8052 firmy Intel.


Model ten wybrano, ponieważ jako jedyny posiadał wystarczającą ilość pamięci danych do przechowywania komunikatów.


Schemat urządzenia przedstawiony został na rys 1.





Wykorzystano tu typową aplikacje mikrokontrolera w konfiguracji z wewnętrzną pamięcią programu. Dla zapewnienia poprawnej pracy interfejsu szeregowego system jest taktowany zegarem 11,059MHz. (przy innej częstotliwości oscylatora np.: 12MHz nie można uzyskać typowych prędkości transmisji)


Wyświetlacz LCD jest samodzielnym zamkniętym modułem integrującym w sobie matrycę LCD i układy sterujące zobrazowaniem, komunikującym się z otoczeniem poprzez 4 lub 8- bitową magistralę danych i 3 linie sterujące.


Do sterowania wyświetlaczem LCD wykorzystano: port P0 jako dwukierunkową magistralę danych, linie P3.2, P3.6 i P3.7 jako linie sterujące. 


Potencjometr R2 służy do regulacji kontrastu wyświetlacza.





Do obsługi klawiatury wykorzystano port P1. Naciśnięcie klawisza jest wykrywane jako pojawienie się logicznego zera na linii portu odpowiadającej danemu klawiszowi. Drabinka rezystorowa wymusza na liniach portu stan wysoki w przypadku zwolnienia klawisza,


Przy problemach ze stabilnym odczytem stanu klawiszy można dołączyć pomiędzy końcówki przełączników kondensatory SMD o pojemności 470-1000nF.





Układ MAX232 pełni funkcje konwertera poziomów napięć z zakresu TTL do poziomów akceptowanych przez RS232C. Linie RxD i TxD służą do przesyłania danych do i od konsoli. Linia CTS konsoli wykorzystywana jest do zerowania konsoli przez system nadrzędny.





Elementy C2, R3 zapewniają zerowanie mikrokontrolera przy włączeniu zasilania. Tranzystor Q1 wykorzystywany jest  do zerowania konsoli na żądanie z zewnątrz linią CTS konsoli.


Elementy R10, R11 DZ1, DZ2 zabezpieczają konsolę przed odwrotnym podłączeniem napięć zasilających. 





Urządzenie wymaga podłączenia 2 stabilizowanych napięć zasilających:


 +5V (300mA) dla konsoli 


 -5V (10mA)  dla polaryzacji wyświetlacza.





Układ U3 będzie montowany opcjonalnie w zależności od potrzeb. Może to być pamięć RAM lub EEPROM z interfejsem I2C (np. AT24C64)

Rys 1. Schemat elektryczny konsoli


�
Oprogramowanie mikrokontrolera





Przystępując do konstrukcji oprogramowania dla mikrokontrolera przyjęto następujące  założenia:


Ustawienia interfejsu szeregowego:  1200,n,8,1,e (prędkość 1200, bez kontroli parzystości, 8 bitowe znaki, 1 bit stopu);


Komunikaty przesyłane są  w formacie ASCII;


Każdy przesyłany komunikat zakończony jest znakiem #00h;


Adres IP przesyłany jest w postaci: xxx.xxx.xxx.xxx/xx


W przypadku braku naturalnej aktywności ze strony konsoli, co pewien czas będzie wysyłany komunikat NOP w celu wykrycia ewentualnej awarii konsoli lub oprogramowania zarządzającego.


Zdefiniowano komunikaty:





Rozkaz�
Kierunek�
Opis�
�
I�
uC->PC�
żądanie inicjalizacji konsoli�
�
A”adres_IP”�
uC<-PC�
Przesłanie konsoli aktualnego adresu IP�
�
A”adres_IP”�
uC->PC�
Potwierdzenie aktualnego adresu IP �
�
A�
uC->PC�
Prośba o podanie aktualnego adresu IP przy pomocy polecenia A”adres_IP”�
�
Dn�
uC<-PC�
Załaduj n komunikatów


Ustawienie ilości przesyłanych komunikatów do konsoli�
�
Rn�
uC->PC�
Nadaj n-ty komunikat


żądanie przesłania do konsoli n-tego komunikatu�
�
Mn”tresc_komunikatu”�
uC<-PC�
przesłanie do konsoli treści n-tego komunikatu�
�
Mn”tresc_komunikatu”�
uC->PC�
Potwierdzenie odebrania n-tego komunikatu przez konsole�
�
B�
uC->PC�
Zakończenie inicjalizacji i rozpoczęcie normalnej pracy konsoli�
�
N�
uC<-PC


uC->PC�
NOP. Nic nie rób.


Wykorzystywane do sprawdzania poprawnej pracy konsoli i oprogramowania obsługującego. W przypadku braku aktywności co pewien ustalony czas konsola wysyła komunikat NOP do PC. Jeśli PC nie odpowie w założonym czasie wyświetlany jest komunikat o awarii oprogramowania obsługującego konsole.


Jeśli konsola nie prześle  w ustalonym czasie komunikatu NOP, oprogramowanie obsługujące stwierdzi zatrzymanie konsoli i rozpocznie automatycznie procedurę inicjalizacji konsoli. �
�
En�
uC->PC�
Polecenie wykonania zadania związanego z n-tą pozycją w menu konsoli. �
�
E0”adres_IP”�
uC->PC�
Szczególny przypadek komunikatu E wykorzystywany do zmiany aktualnego adresu IP .�
�
Rn”tresc_odpowiedzi”�
uC<-PC�
Przesłanie rezultatów wykonania polecenia żądanego komunikatem En. Treść odpowiedzi nie może być dłuższa niż 48 znaków. Elementem formatującym jest znak LF (#0Ah) oznaczający przejście do następnej linii.�
�
Rn”tresc_odpowiedzi”�
uC->PC�
Potwierdzenie otrzymania rezultatów wykonania polecenia n.�
�



Po włączeniu zasilania lub zerowaniu konsoli wykonywane są procedury inicjalizujące wyświetlacz LCD, układ czasowo - licznikowy, port szeregowy i system przerwań.


Następnie wysyłane jest do systemu nadrzędnego żądanie inicjalizacji konsoli ( I ).


Na ekranie konsoli wyświetlany jest komunikat:





Initializing...





W odpowiedzi na żądanie inicjalizacji system nadrzędny przesyła informacje o aktualnym�adresie IP (A). Konsola potwierdza odebranie odsyłając komunikat (A), i oczekuje na informacje o ilości pozycji do załadowania (D).  Jakiekolwiek błędy na tym etapie powodują rozpoczęcie procesu inicjalizacji.


Następnie konsola wysyła  żądanie przesłania treści kolejnych pozycji (R).


W odpowiedzi system nadrzędny przesyła treść pozycji (M). Konsola potwierdza odebranie odsyłają komunikat do systemu nadrzędnego (M).


Po załadowaniu wszystkich pozycji konsola wysyła komunikat o starcie systemu (B) i wyświetla komunikat powitalny na ekranie LCD:





Abng Poller


(c) 1999 DSRG AGH





Od tej chwili konsola oczekuje na akcje z klawiatury. W przypadku braku aktywności co określony czas wysyłany jest komunikat NOP, na który system nadrzędny odpowiada komunikatem NOP. W przypadku braku w ustalonym czasie odpowiedzi od systemu nadrzędnego wyświetlany jest komunikat:


 Daemon failure.





W tym momencie Konsola reaguje tylko na naciśnięcie klawisza HELP (wyświetlana jest krótka informacja o sposobie korzystania z systemu), lub MENU.


Naciśnięcie klawisza MENU powoduje pojawienie się na wyświetlaczu następującej informacji:


    MENU    ->


Change IP Addr


Zgodnie z ustaleniami na pierwszej pozycji znajduje się zawsze polecenie zmiany adresu IP.


Pozostałe polecenia mogą być dowolnie konfigurowane przez system nadrzędny.


Wyświetlana w prawym lub lewym górnym rogu strzałka informuje o dostępności innych poleceń w MENU. Brak strzałki w lewo oznacza, że wyświetlane jest pierwsze dostępne polecenie MENU, brak strzałki w prawo, oznacza ostatnie dostępne polecenie. 


Za pomocą klawiszy PREV i NEXT możemy poruszać się po poleceniach MENU.


Naciśnięcie klawisza UP powoduje przejście do pierwszego polecenia w MENU, a klawisza DOWN do ostatniego polecenia.


Naciśnięcie klawisza CANCEL powoduje powrót do ekranu powitalnego, a klawisza ENTER wykonanie polecenia MENU wyświetlanego na ekranie.





W przypadku poleceń ładowanych z systemu nadrzędnego konsola wysyła komunikat En i przechodzi do oczekiwania na odpowiedź (R).


Po otrzymaniu odpowiedzi konsola potwierdza odbiór komunikatu (R), i wyświetla wynik wykonania polecenia. I oczekuje na naciśnięcie dowolnego klawisza.


�
W przypadku wykonania polecenia Change IP Addr wyświetlana jest prośba o podanie adresu IP.





Na ekranie pojawia się następująca informacja:


Change IP Addr


127.000.000.001


gdzie adres: 127.000.000.001 jest adresem przesłanym do konsoli w trakcie inicjalizacji.


Na pierwszej z lewej pozycji adresu  pojawia się kursor w postaci  migającego znaku ”_ ” . Kursor wskazuje pozycje adresu którą w danej chwili można modyfikować.


Zmiany wartości pozycji dokonuje się klawiszami UP i DOWN. Klawisze PREV i NEXT służą do zmiany pozycji którą chcemy modyfikować. Naciśnięcie klawisza ENTER  wyświetla prośbę o podanie maski.:


Change IP Addr


127.000.000.001


Mask: 24     





wyświetlana wartość maski jest tą jaka została pobrana w trakcie inicjalizacji konsoli.


Modyfikacji dokonuje się w taki sam sposób jak przy wprowadzaniu adresu.


Naciśnięcie ENTER kończy  wprowadzanie nowych ustawień IP. Poprawność adresu nie jest sprawdzana. Wprowadzony adres jest wysyłany w komunikacie E0”127.000.000.001/24”.


System nadrzędny sprawdza poprawność adresu i informuje komunikatem R o wynikach zmiany.





Naciśnięcie klawisza CANCEL powoduje powrót do menu bez dokonywania zmian i wysyłania komunikatów.





W celu zapewnienia koherencji danych w konsoli z danymi w systemie nadrzędnym po wykonaniu dowolnego polecenia i przekazaniu rezultatów następuje procedura pobrania bieżącego adresu IP z systemu nadrzędnego.





Nie została zaimplementowana pełna obsługa klawisza HELP.











Program obsługi konsoli został napisany w asemblerze 8051 i skompilowany z użyciem crossasemblera firmy MetaLink w wersji 1.2h . Kod źródłowy został załączony w dodatku C tego opracowania.


�
Wnioski 





Po zakończeniu prac nad konsolą nasuwają się następujące wnioski:





Zmiana sposobu obsługi klawiatury wymagająca pewnej ingerencji w sprzęt tzn. Zastąpienie istniejącej klawiatury przez klawiaturę matrycową. Poprawi to stabilność odczyt klawiszy.


Zorientowanie oprogramowania obsługującego na obsługę zdarzeń i pośrednictwo w bezpośrednim dostępie do wyświetlacza LCD ( konsola obsługiwałaby zdarzenie typu „naciśnięto klawisz” oraz kilka funkcji dających bezpośredni dostęp do wyświetlacza (wypisz znak, wyślij rozkaz, odczytaj status, odczytaj dane). Uprości to oprogramowanie konsoli, przy jednoczesnym zwiększeniu jej uniwersalności. Ceną będzie znaczne skomplikowanie oprogramowania systemu nadrzędnego.








�

�
Załącznik A


Wykaz elementów konsoli








Symbol�
Nazwa elementu�
ilość sztuk�
Uwagi�
�
U1�
AT89C52�
1�
�
�
U2�
MAX232	�
1�
�
�
U3�
AT24C64�
1�
element montowany opcjonalnie�
�
Q1�
BC 177�
1�
lub inny typu PNP�
�
DZ1,DZ2�
Dioda Zenera na 5,1 V 


np. BZP680C5V1�
2�
�
�
Y1�
Kwarc 11.059 MHz�
1�
�
�
C1,C3�
Kondensator 33pF�
2�
�
�
C2,C4,C5,


C6,C7�
Kondensator 10 (F�
5�
�
�
C8�
Kondensator 100 (F�
1�
�
�
C10-C12�
Kondensator SMD 100 nF�
3�
nie zaznaczone na schemacie ideowym �
�
C20-C27 *�
Kondensator SMD 470 nF�
8�
nie zaznaczone na schemacie ideowym�
�
DR1�
Drabinka rezystorowa 1k( x 8�
1�
�
�
R1,R2,R3�
Rezystor 10 k(�
3�
�
�
R5,R6�
Rezystor 5,1 k(�
2�
�
�
R10,R11�
Rezystor 10 (�
2�
�
�
SW1-SW8�
Mikroprzełączniki 10mm�
8�
�
�
J1�
Złącze igłowe 2x7 �
1�
�
�
J2�
Złącze igłowe 1x3 �
1�
�
�
J3�
Złącze igłowe 1x4 �
1�
�
�



* Kondensatory C20-C27 montowane opcjonalnie pomiędzy styki mikroprzełączników w celu tłumienia drgań zestyków i poprawienia stabilności pracy klawiatury konsoli.�
Załącznik B


Rozmieszczenie elementów na płytce drukowanej





Do opracowania załączone są kliszy z mozaiką ścieżek płytki drukowanej. Ze względów technologicznych płytka została powielona dwukrotnie.



















































































Elementy J1, C10, S1 - S8 należy zamontować po stronie druku.


Dodatkowe kondensatory C20 - C27 montuje się po stronie elementów pomiędzy wyprowadzenie mikroprzełączników.


�
Załącznik C


Kod źródłowy programu konsoli








;		 Konsola LCD dla systemu ABng


;		(c) 1999 ZAB for DSRG DCS AGH


;


; KOMUNIKATY MENU ZEBRANE W FORMIE TABLICY OD ADRESU MSG_BASE=90H


; KAZDY KOMUNIKAT MA MAKSYMALNIE 16 ZNAKOW BEZ ZNAKU KONCA


; KOMUNIKAT KROTSZY DOPELNIC SPACJAMI


; MAKSYMALNIE MOZE BYC 9 KOMUNIKATOW


; DA TO W SUMIE MAKSYMALNIE 10 POZYCJI W MENU


; NIE ZOSTALA ZAIMPLEMENTOWANA OBSLUGA INTERFEJSU I2C





$MOD52


$OBJECT


$NOPAGING


;==================================================================


; DEFINICJE ROZKAZOW WYSWIETLACZA LCD DLA PROCEDURY WRITE_COMMAND





CLEAR_DISP	EQU 01H


HOME		EQU 02H


SHOW_CUR	EQU 0FH


HIDE_CUR	EQU 0CH


;==================================================================





LCD_RS		EQU P3.7	; linia sterujaca wyswietlacza LCD 


LCD_E		EQU P3.2	; linia sterujaca wyswietlacza LCD


LCD_RW		EQU P3.6	; linia sterujaca wyswietlacza LCD


SDA		EQU P3.5 	; opcjonalna linia SDA magistrali IIC


SCL		EQU P3.4 	; opcjonalna linia SCL magistrali IIC


 


WD		EQU 2AH		; licznik WATCHDOG-a


RS_FLAG		EQU 2CH		; flaga transmisji przez RS


C_FLAG		EQU 2DH		; znaczniki dla konsoli


NR_COMMAND	EQU 2EH		; ilosc zaladowanych komunikatow


RS_PTR		EQU 2FH		; wskaznik do aktualnej pozycji w buforze 						 	; danych do transmisji


RS_BUF		EQU 30H		; bufor wejsciowy RS


IP_BUF		EQU 70H		; miejsce na adres IP


TMP_IP		EQU 3AH		; tymczasowe przechownie adresu IP przy zminaie ustawien


MSG_BASE	EQU 90H		; poczatek pamieci na komunikaty





RS_ACTIVE 	EQU RS_FLAG.7		; wskaznik aktywnosci na RS232


RS_TIMEOUT	EQU RS_FLAG.6		; znacznik przeterminowania


NOP_SENT	EQU RS_FLAG.5		; znaczni wyslania NOP


MSG_OK		EQU RS_FLAG.0		; wskaznik odebrania calego komunikatu


MSG_SENT	EQU RS_FLAG.1		; wskaznik zakonczenia wysylania komunikatu





IP_OK		EQU C_FLAG.7		; wskaznik zakonczenia wprowadzania IP


MASK_OK		EQU C_FLAG.6		; wskaznik zakonczenia wprowadzania maski


IP_CANC		EQU C_FLAG.0		; wnacisniecia CANCEL przy wprowadzanu IP 





MAX_COMMAND	EQU 07H		; maksymalna ilosc przyjmowanych komunikatow	


MAX_TIME	EQU 0FEH	; czas przeterminowania (5s = 46h)


LF		EQU 0AH		; znak przejscia do nastepnej linii	





;============================================================================


;DEFINICJE SYMBOLI KLAWISZY DLA SPAWACZA





KEY_NEXT	EQU ACC.0	


KEY_PREV	EQU ACC.6	


KEY_UP		EQU ACC.2


KEY_DOWN	EQU ACC.4


KEY_HELP	EQU ACC.5


KEY_CANCEL	EQU ACC.3


KEY_ENTER	EQU ACC.1


KEY_MENU	EQU ACC.7





;============================================================================


; DEFINICJE WYSYLANYCH KOMEND


USER_COMMAND	EQU 'E'


INIT_REQ	EQU 'I'


REQ_MSG		EQU 'R'


BEGIN_MSG	EQU 'B'


REPLY_MSG	EQU 'R'


NOP_MSG		EQU 'N'


SET_IP		EQU 'A'


;============================================================================





	ORG 0000H


	AJMP START





;obsluga przerwania INT0


	ORG 0003H


	RETI





;obsluga przerwania licznika T0


	ORG 000BH


	RETI





;obsluga przerwania INT1


	ORG 0013H


	RETI





;obsluga przerwania licznika T1


	ORG 01BH


	RETI








;obsluga przerwania portu szeregowego


	ORG 0023H


	LJMP RS_INT	





;obsluga przerwania licznika T2


	ORG 02BH


	LJMP T2_SERVICE


;	RETI





	ORG 0100H





; procedury incjalizujace 


START:	


	MOV C_FLAG,#00H			; wyzerowanie znacznikow konsoli


	MOV RS_FLAG,#00H		; wyzerowanie wskaznikow dla RS


	MOV NR_COMMAND,#00H		; wartosc poczatkowa ilosci dostepnych polecen





	LCALL INIT_LCD	 		; inicjalizacja wyswietlacza


	LCALL INIT_RS			; inicjalizacja interfejsu szeregowego


	LCALL INIT_T2			; inicjalizacja licznika T2





	MOV A,#CLEAR_DISP


	LCALL WRITE_COMMAND


	LCALL DELAY


	MOV A,#HIDE_CUR


	LCALL WRITE_COMMAND


	MOV C_FLAG,#00H			; wyzerowanie znacznikow konsoli


	MOV RS_FLAG,#00H		; wyzerowanie wskaznikow dla RS


	MOV R0,#MSG_BASE





CLR_MSG:				; zapelnienei bufora komunikatow spacjami


	MOV @R0,#20H		


	INC R0


	CJNE R0,#00H,CLR_MSG


	SETB EA				; odblokowanie przerwan


	MOV A,#0C0H


	LCALL WRITE_COMMAND


	MOV DPTR,#WELC3			; wyswietlenie 'Initializing...'


	LCALL PRINT_STR_C





; wyslanie komunkatu incjalizacji panelu


INIT_CON:





	MOV R0,#RS_BUF


	MOV @R0,#INIT_REQ


	INC R0


	MOV @R0,#00H


	MOV R0,#RS_BUF


	LCALL WRITE_RS





; oczekiwanie na odpowiedz demona


DAEMON_WAIT:


	JB RS_TIMEOUT,START		; po przekroczeniu czasu oczekiwania od poczatku


	JNB MSG_OK,DAEMON_WAIT





	CLR MSG_OK





; czekamy na A"IP/MASK"





	MOV R0,#RS_BUF


	MOV A,@R0


	CJNE A,#'A', START		; jesli odebrany komunikat to nie a to od nowa


	INC R0


; kopiujemy caly komunikat do wykorzystania 


; zmieniamy tylko A na E0


	MOV R1,#IP_BUF


	MOV @R1,#USER_COMMAND


	INC R1


	MOV @R1,#'0'


	INC R1


COPY_IP:


	MOV A,@R0


	MOV @R1,A	


	INC R0


	INC R1


	CJNE A,#00H,COPY_IP


; odsylamy echo 


	MOV R0,#RS_BUF


	LCALL WRITE_RS





; i czekamy na demona


	CLR MSG_OK


WAIT1:


	JB RS_TIMEOUT,START		; w przypadku bledu skok do poczatku


	JNB MSG_OK,WAIT1





; teraz oczekujemy na 'Dn'


	


	MOV R0,#RS_BUF


	MOV A,@R0


	CJNE A,#'D', START		; jesli odebrany komunikat to nie D to od nowa


	INC R0


	MOV A,@R0			; ustawiamy ilosc ladowanych polecen


	CLR C


	SUBB A, #'0'			; zamiana kodu ASCII na cyfre 0-9


	MOV NR_COMMAND,A		; zapamietanie ilosci ladowanych komunikatow





; sprawdzamy ,czy wystarczy pamieci na komunikaty


	MOV A,NR_COMMAND


	CLR C


	SUBB A,#MAX_COMMAND	


	JC IC1		


	LJMP START			; proba zaladowania zbyt wielu komunikatow 





; teraz bedziemy pobierac kolejne komunikaty


IC1:


	MOV AR2,#01H			; w R2 bedziemy przechowywac licznik komunikatow





DOWNLOAD_MSG:


; wysylamy zadanie n-tego komunikatu Rn


	MOV R0,#RS_BUF


	MOV @R0,#REQ_MSG


	INC R0


	MOV A,R2


	ADD A,#'0'


	MOV @R0,A


	INC R0


	MOV @R0,#00H


	MOV R0,#RS_BUF


	LCALL WRITE_RS			; wyslanie zadania


	CLR MSG_OK


	JNB MSG_OK,$			; czekamy na odpowiedz demona





; analizujem otrzymana odpowiedz


	CLR MSG_OK


	MOV R0,#RS_BUF


	MOV A,@R0


	CJNE A,#'M',DOWNLOAD_MSG	; jesli nie przyszlo M to od nowa


					; zaczynamy ladowanie komunikatow


	INC R0


	MOV A,@R0


	CLR C


	SUBB A, #'0'


	CJNE A,AR2,DOWNLOAD_MSG		; sprawdzamy czy zgadza sie numer komunikatu 


	INC R0				; nastepny powiniwn byc " , ale nie sprawdzamy


	INC R0				; pierwszy znak komunikatu





; wyznaczamy adres w buforze komunikatow (n-1)*16+msg_base


	DEC A


	RL A


	RL A


	RL A


	RL A


	ADD A,#MSG_BASE


	MOV R1,A			; do R1 adres przeznaczenia





; kopiujemy komunikat 


; wazne aby komunikaty ladowac po kolei !!!





CP_MSG:	


	MOV A,@R0			; pierwszy znak komunikatu


	MOV @R1,A			; do bufora komunikatow


	INC R0


	INC R1


	CJNE A,#'"',CP_MSG		; jesli cudzyslow, to koniec komuniaktu


	DEC R1


	MOV @R1,#20H





; komunikat zapamietany, odsylamy echo odebranego komunikatu


	MOV R0,#RS_BUF


	LCALL WRITE_RS





; mala petla opozniajaca


	PUSH AR1


	PUSH AR2


	MOV R1,#0FFH


QAE1:	MOV R2,#07FH


	DJNZ R2,$


	DJNZ R1,QAE1


	POP AR2


	POP AR1





	MOV A,R2


	CLR C


	SUBB A,NR_COMMAND


	JZ ALL_MSG


	INC R2


	SJMP DOWNLOAD_MSG


; zaladowano wszystkie komunikaty





ALL_MSG:


;wysylamy komunikat o startcie konsoli


	MOV R0,#RS_BUF


	MOV @R0,#BEGIN_MSG


	INC R0


	MOV @R0,#00H


	MOV R0,#RS_BUF


	LCALL WRITE_RS





; wyswietlenie komunikatu powitalnego	


PETLA:


	MOV A,#CLEAR_DISP


	LCALL WRITE_COMMAND


	MOV A,#0C0H


	LCALL WRITE_COMMAND	


	MOV DPTR,#WELC1


	LCALL PRINT_STR_C


	MOV A,#090H


	LCALL WRITE_COMMAND


	MOV DPTR,#WELC2	


	LCALL PRINT_STR_C	


MAIN:


	MOV A,#0D0H


	LCALL WRITE_COMMAND





; po wyswietlenu ekranu powitalnego czekamy na nacisniecie klawisza	


	LCALL READKB


	JB KEY_MENU, MENU


	JB KEY_HELP, HELP1





	SJMP MAIN		; nic nie nacisnieto oczekujemy dalej





;===========================================================================	


; obsluga nacisniecia klawisza HELP





HELP1:


	MOV A,#CLEAR_DISP


	LCALL WRITE_COMMAND


	MOV DPTR,#HELPM1


	LCALL PRINT_STR_C


	MOV A,#0C0H


	LCALL WRITE_COMMAND


	MOV DPTR,#HELPM2


	LCALL PRINT_STR_C


	MOV A,#090H


	LCALL WRITE_COMMAND


	MOV DPTR,#HELPM3


	LCALL PRINT_STR_C


	MOV A,#0D0H


	LCALL WRITE_COMMAND


	MOV DPTR,#MSG2


	LCALL PRINT_STR_C


HLP1_1:	LCALL READKB


	JZ HLP1_1


	SJMP PETLA


;===========================================================================	





MENU:


; obsluga nacisniecia klawisza MENU





;	next : pozycja+1, skok do wyswietl_pozycje


;	prev : pozycja-1, skok do wyswietl_pozycje


;	up : pozycja=0, skok do wyswietl_pozycje


;	down: pozycja=ostatnia, skok do wyswietl_pozycje


;	cancel : wyswietl komunikat powitalny


;	enter : wykonaj pozyje


;	help : wyswietl help








; czyszczenie wyswietlacza


	MOV A,#CLEAR_DISP


	LCALL WRITE_COMMAND


; napis Menu w 1 linii na srodku


	MOV DPTR, #MSG1


	LCALL PRINT_STR_C





; biezaca pozycja menu - UserCommand(0) zmiana IP


	MOV R3,#00H		;w R3 biezaca pozycja





; przestawienie kursora na poczatek 2 linii


MENU_REFRESH:





	MOV A,#0C0H


	LCALL WRITE_COMMAND





; wyswietlenie biezacej pozycji menu (PARAMETR W R3)





	LCALL DISP_COMMAND





;jesli pozycja == 0 wyswietl tylko -> 


MENU_ONE:


; jesli nie zaladowano zadnych komunikatow


	MOV A,NR_COMMAND


	JNZ MENU_FIRST


	MOV A,#80H


	LCALL WRITE_COMMAND


	MOV R7,#' '


	LCALL PRINT


	MOV A,#8FH


	LCALL WRITE_COMMAND


	MOV R7,#' '


	LCALL PRINT


	SJMP MENU_KEY


	


MENU_FIRST:


	MOV A,R3


	JNZ MENU_LAST


; kasujemy znak wyswietlany po lewej stronie 


	MOV A,#80H


	LCALL WRITE_COMMAND


	MOV R7,#' '


	LCALL PRINT


	MOV A,#8FH


	LCALL WRITE_COMMAND


	MOV R7,#7EH


	LCALL PRINT


	SJMP MENU_KEY





;jesli biezaca pozycja == ostatnia wyswietl tylko <-





MENU_LAST:


	MOV A,NR_COMMAND


	CJNE A,AR3,MENU_NORM


	MOV A,#80H


	LCALL WRITE_COMMAND


	MOV R7,#7FH


	LCALL PRINT


	MOV A,#8FH


	LCALL WRITE_COMMAND


	MOV R7,#20H


	LCALL PRINT


	SJMP MENU_KEY





; pozycja>0 && pozycja<ostatnia wyswietl <- i ->





MENU_NORM:


	MOV A,#80H


	LCALL WRITE_COMMAND


	MOV R7,#7FH


	LCALL PRINT


	MOV A,#8FH


	LCALL WRITE_COMMAND


	MOV R7,#7EH


	LCALL PRINT





;sprawdz czy nacisiety klawisz





MENU_KEY:


	LCALL READKB


; nacisniety next: POZYCJA+1


	JNB KEY_NEXT, MKR1


; if pozycja==nr_command nie zwiekszamy


	MOV A,NR_COMMAND


	CLR C


	SUBB A,R3


	JZ MENU_REFRESH


	INC R3	


	AJMP MENU_REFRESH





; nacisniety prev POZYCJA-1


MKR1:


	JNB KEY_PREV,MKR2


	MOV A,R3


	JZ MENU_REFRESH


	DEC R3


	AJMP MENU_REFRESH





; nacisniety up POZYCJA=0


MKR2:


	JNB KEY_UP, MKR3


	MOV R3,#00H


	AJMP MENU_REFRESH





; nacisniety down POZYCJA=NR_COMMAND


MKR3:


	JNB KEY_DOWN, MKR4


	MOV R3,NR_COMMAND


	AJMP MENU_REFRESH





; nacisniety CANCEL


MKR4:	


	JNB KEY_CANCEL, MKR5


	AJMP PETLA





; nacisniety HELP


MKR5:


	JNB KEY_HELP, MKR6


;	LCALL DISP_HELP





; nacisniety ENTER 


MKR6:	JNB KEY_ENTER, MKR7


	LCALL RUN_COMMAND	





	AJMP MENU		; po wykonaniu polecenia wyswietlamy Menu





MKR7:		


;nic nie nacisnieto


	SJMP MENU_KEY	





;KONIEC: LJMP M0





;============================================================================





RUN_COMMAND:


; wykonanie wybranej pozycji menu


; w r3 przekazywany numer wybranej pozycji


; jesli r3==0 to pobierz adres


; 


; umiesc w buforze mnemonik polecenia 


; umiesc kod polecenia 


; wyslij komunikat z bufora


; czekaj na odpowiedz


; wyswietl odpowiedz


	


	MOV A,R3


	JNZ NO_IP


	CLR IP_CANC


	CLR IP_OK


	CLR MASK_OK


	LCALL GET_IP			; w IP_BUF ma byc gotowe polecenie do wyslania


					; ewentualnie ustawiona flaga IP_CANC


	JB IP_CANC,R_EXIT


	MOV R0,#IP_BUF	


	SJMP R_SEND			; wyslanie komunikatu


NO_IP:	


; ustawienie na pocztek bufora 


	MOV R0,#RS_BUF


	MOV A,#USER_COMMAND


	MOV @R0,A


	INC R0


; numer polecenia z r3 zmieniamy na ASCII (od 0 do 9) 


	MOV A,R3


	ADD A,#'0'


	MOV @R0,A


	INC R0


	CLR A


	MOV @R0,A


;i wysylamy przez RS





	MOV R0,#RS_BUF


R_SEND:


	LCALL WRITE_RS


R_ERR:	CLR MSG_OK


; teraz czekamy na odpowiedz	





	MOV A,#CLEAR_DISP


	LCALL WRITE_COMMAND


	JNB MSG_OK,$





; sprawdzamy czy przyszlo to co trzeba





	MOV R1,#RS_BUF


	MOV A,@R1


	CJNE A,#REPLY_MSG,R_ERR		; REPLY_COMMAND


	INC R1


	MOV A,@R1


	CLR C


	SUBB A,#'0'


	CJNE A,AR3,R_ERR		; nr komunikatu





; i na wyswietlacz





	MOV R1,#RS_BUF+3


	LCALL PRINT_RESULT





R_PR_EXIT:


	MOV A,#0D0H


	LCALL WRITE_COMMAND


	MOV DPTR,#MSG2


	LCALL PRINT_STR_C





; i odsylamy potwierdzenie


	MOV R0,#RS_BUF


	LCALL WRITE_RS





; Uzgodnienie adresu IP


	LCALL DELAY			; male opoznienie


	MOV R0,#RS_BUF


	MOV A,#SET_IP			; wyslanie polecenia A


	MOV @R0,A


	INC R0


	CLR A


	MOV @R0,A





	MOV R0,#RS_BUF


	LCALL WRITE_RS





R_IP_ERR:	


	CLR MSG_OK


	JNB MSG_OK,$


; sprawdzamy czy jest A"..."





	MOV R1,#RS_BUF


	MOV A,@R1


	CJNE A,#SET_IP,	R_IP_ERR	; czekaj na przyjscie adresu


	INC R1


	INC R1





; kopiujemy adres


	MOV R0,#IP_BUF+3


	MOV R7,#17





R_IP_CP:


	MOV A,@R1


	MOV @R0,A


	INC R0


	INC R1


	DJNZ R7, R_IP_CP


	


; odeslanie potwierdzenia


	MOV R0,#RS_BUF


	LCALL WRITE_RS


; czekamy na nacisniecie klawisza


KQWA:	LCALL READKB


	JZ KQWA


	MOV A,#CLEAR_DISP


	LCALL WRITE_COMMAND





	


R_EXIT:


	RET





;============================================================================





; Pobranie adresu IP z konsoli





GET_IP:	


; czyszczenie wyswietlacza


	MOV A,#CLEAR_DISP


	LCALL WRITE_COMMAND


	MOV DPTR,#IPMSG1


	LCALL PRINT_STR_C


	MOV A,#0C0H


	LCALL WRITE_COMMAND


	MOV A,#SHOW_CUR


	LCALL WRITE_COMMAND


; wyswietliamy adres z IP_BUF i kopiujemy do RS_BUF


	MOV R1,#0FH


	MOV R0,#IP_BUF+3


CP_IP0:


	MOV A,@R0


	MOV R7,A


	LCALL PRINT


	INC R0


	DJNZ R1,CP_IP0





	MOV A,#0C0H


	LCALL WRITE_COMMAND


; wypelnienie bufora wlasciwa zawartoscia


; do TMP_IP przerzucimy tylko sam adres ip+maska bez cudzyslowow





	MOV R0,#IP_BUF+3		; pierwszy znak adresu 


	MOV R1,#TMP_IP


CP_IP1:


	MOV A,@R0


	MOV @R1,A


	INC R0


	INC R1


	CJNE A,#'"',CP_IP1		; jak " to juz caly adres 





; wyswietl znak na biezacej pozycji





	MOV R4,#00H


GIP0:	LCALL READKB


; nacisniety NEXT: nastepna pozycja


	JNB KEY_NEXT, GIP1


	MOV A,R4			; wskaznik do A


	INC A				; nastepna pozycja


	CJNE A,#03H,GIP01		; pierwsza kropka 


	INC A			


GIP01:	CJNE A,#07H,GIP02		; druga kropka


	INC A


GIP02:	CJNE A,#0BH,GIP03		; trzecia kropka


	INC A


GIP03:	CJNE A,#0FH,GIP04		; koniec 


	DEC A				; zmniejszamy aby kursor zostal na koncu


GIP04:		


	MOV R4,A			; zapamietanie w R4 nowej pozycji


; ustaw kursor na nowej pozycjii


	ADD A,#0C0H


	LCALL WRITE_COMMAND


; wyswietl znak pod kursorem


	MOV A,R4


	ADD A,#TMP_IP


	MOV R0,A


	MOV AR7,@R0			; do r7 zawartosc bufora


	LCALL PRINT


;przesuniecie kursora na wlasciwa pozycje


	MOV A,R4


	ADD A,#0C0H


	LCALL WRITE_COMMAND





	AJMP GIP0





; nacisniety PREV: poprzednia pozycja


GIP1:


	JNB KEY_PREV,GIP2


	MOV A,R4			; wskaznik do A


	DEC A				; poprzednia pozycja


	CJNE A,#03H,GIP11		; pierwsza kropka 


	DEC A			


GIP11:	CJNE A,#07H,GIP12		; druga kropka


	DEC A


GIP12:	CJNE A,#0BH,GIP13		; trzecia kropka


	DEC A


GIP13:	CJNE A,#0FFH,GIP14		; koniec 


	INC A				; zmniejszamy aby kursor zostal na koncu


GIP14:		


	MOV R4,A			; zapamietanie w R4 nowej pozycji





; ustaw kursor na nowej pozycjii


	ADD A,#0C0H


	LCALL WRITE_COMMAND


; wyswietl znak pod kursorem


	MOV A,R4


	ADD A,#TMP_IP


	MOV R0,A


	MOV AR7,@R0


	LCALL PRINT


;przesuniecie kursora na wlasciwa pozycje


	MOV A,R4


	ADD A, #0C0H


	LCALL WRITE_COMMAND


	


	AJMP GIP0


	


; nacisniety UP: zwieksz pozycje o 1


GIP2:


	JNB KEY_UP, GIP3


	MOV A,R4


	ADD A,#TMP_IP


	MOV R0,A


	MOV A,@R0


	CLR C


	SUBB A,#'0'			; zamiana ASCII na cyfre


	INC A


	CJNE A,#0AH,GIP21		; jesli 10 to 0


	CLR A


GIP21:	ADD A,#'0'


	MOV @R0,A


	MOV R7,A


	LCALL PRINT


	MOV A,R4


	ADD A, #0C0H


	LCALL WRITE_COMMAND


	AJMP GIP0	


		


; nacisniety DOWN: zmniejsz pozycje o 1


GIP3:


	JNB KEY_DOWN, GIP4


	MOV A,R4


	ADD A,#TMP_IP


	MOV R0,A


	MOV A,@R0


	CLR C


	SUBB A,#'0'			; zamiana ASCII na cyfre


	DEC A


	CJNE A,#0FFH,GIP31


	MOV A,#09H


GIP31:	ADD A,#'0'


	MOV @R0,A


	MOV R7,A


	LCALL PRINT


	MOV A,R4


	ADD A, #0C0H


	LCALL WRITE_COMMAND


	AJMP GIP0


; nacisniety CANCEL


GIP4:	


	JNB KEY_CANCEL, GIP5


	SETB IP_CANC			; ustawienie znacznika


	AJMP GIP_EXIT	





; nacisniety HELP


GIP5:


	JNB KEY_HELP, GIP6


;	LCALL DISP_HELP





; nacisniety ENTER 


GIP6:	JNB KEY_ENTER, GIP7


	SETB IP_OK


	SJMP GIP8


GIP7:		


;nic nie nacisnieto


	AJMP GIP0


; teraz pobierzemy maske





GIP8:


	MOV A,#90H


	LCALL WRITE_COMMAND		; ustawienie kursora na poczatek 3 linii


	MOV DPTR,#IPMSG3


	LCALL PRINT_STR_C


	MOV A,#96H


	LCALL WRITE_COMMAND


; wyswietlamy biezaca maske


	MOV R0,#TMP_IP+10H


	MOV AR7,@R0


	LCALL PRINT


	INC R0


	MOV AR7,@R0


	LCALL PRINT


; ustawienie kursora na wlasciwej pozyji


	MOV A,#96h


	LCALL WRITE_COMMAND





	MOV R4,#00


GIPM1:	LCALL READKB





; nacisniety UP


GIPM2:


	JNB KEY_UP, GIPM3


	MOV A,R4


	ADD A,#TMP_IP+10H


	MOV R0,A


	MOV A,@R0


	CLR C


	SUBB A,#'0'			; zamiana ASCII na cyfre


	INC A


	CJNE A,#0AH,GIPM21		; jesli 10 to 0


	CLR A


GIPM21:	ADD A,#'0'


	MOV @R0,A


	MOV R7,A


	LCALL PRINT


	MOV A,R4


	ADD A, #96H


	LCALL WRITE_COMMAND


	AJMP GIPM1	


		


; nacisniety DOWN


GIPM3:


	JNB KEY_DOWN,GIPM4


	MOV A,R4


	ADD A,#TMP_IP+10H


	MOV R0,A


	MOV A,@R0


	CLR C


	SUBB A,#'0'			; zamiana ASCII na cyfre


	DEC A


	CJNE A,#0FFH,GIPM31


	MOV A,#09H


GIPM31:	ADD A,#'0'


	MOV @R0,A


	MOV R7,A


	LCALL PRINT


	MOV A,R4


	ADD A,#96H


	LCALL WRITE_COMMAND


	AJMP GIPM1





GIPM4:


; nacisniety NEXT


	JNB KEY_NEXT,GIPM5


	MOV A,R4


	INC A


	CJNE A,#02,GIPM51


	DEC A


GIPM51:	


	MOV R4,A


	ADD A,#96H


	LCALL WRITE_COMMAND


	AJMP GIPM1


GIPM5:


;nacisniety PREV


	JNB KEY_PREV,GIPM6


	MOV A,R4


	DEC A


	CJNE A,#0FFH,GIPM61


	INC A


GIPM61:


	ADD A,#96H


	LCALL WRITE_COMMAND





	AJMP GIPM1


GIPM6:


; nacisniety CANCEL


	JNB KEY_CANCEL,GIPM7


	SETB IP_CANC


	SJMP GIP_EXIT


	





GIPM7: 


; nacisniety ENTER


	JNB KEY_ENTER,GIPM8


	SETB MASK_OK			; wprowadzona maska


	SJMP GIP_EXIT


GIPM8:


	AJMP GIPM1			; nic nie nacisnieto








GIP_EXIT:


	MOV A,#HIDE_CUR


	LCALL WRITE_COMMAND


; jesli nie bylo cancel to przepiszemy nowa maske do IP_BUF


	JNB IP_CANC,IP_CP10


	RET





IP_CP10:


	MOV R0,#TMP_IP


	MOV R1,#IP_BUF+3


	MOV R6,#13H


IP_CP11:


	MOV A,@R0


	MOV @R1,A


	INC R0


	INC R1


	DJNZ R6,IP_CP11


	RET





;============================================================================





DISP_COMMAND:


;WYSWIETLENIE TEKSTU KOMINIKATU MENU


; numer komunikatu w R3


	MOV A,R3


	JNZ NOT_IP		 


	MOV DPTR,#MENU0			; wyswietlenie komunikatu "Change IP"


	LCALL PRINT_STR_C


	RET





NOT_IP:


; pozostale komunikaty ladowane z PC


; poczatek komunikatu msg_base+((r3-1)*16)


	DEC A


	RL A


	RL A


	RL A


	RL A	


	ADD A,#MSG_BASE


	MOV R0,A			; adres poczatku komunikatu do R0


	MOV R6,#0FH			; licznik znakow


DC1:	MOV A,@R0			; znak do A


	MOV R7,A


	LCALL PRINT			; wyswietl znak


	INC R0


	DJNZ R6,DC1			; i do konca komunikatu


	RET





;============================================================================





READKB:


; obsluga klawiatury na p1


; nacisniety klawisz aktywny zerem (jako wejscie)


; wynik w acc


; zwracamy 00 jesli nic nie nacisnieto





	CLR A


RDP10:	JB P1.0,RDP11


	JNB P1.0,$


	SETB ACC.0


	RET


RDP11:	JB P1.1,RDP12


	JNB P1.1,$


	SETB ACC.1


	RET


RDP12:	JB P1.2,RDP13


	JNB P1.2,$


	SETB ACC.2


	RET


RDP13:	JB P1.3,RDP14


	JNB P1.3,$


	SETB ACC.3


	RET


RDP14:	JB P1.4,RDP15


	JNB P1.4,$


	SETB ACC.4


	RET


RDP15:	JB P1.5,RDP16


	JNB P1.5,$


	SETB ACC.5


	RET


RDP16:	JB P1.6,RDP17


	JNB P1.6,$


	SETB ACC.6


	RET


RDP17:	JB P1.7,RDP18


	JNB P1.7,$


	SETB ACC.7


	RET


RDP18:	CLR A


	MOV R6,#F0


RDP19:	MOV R7,#0FFH


	DJNZ R7,$


	DJNZ R6,RDP19


	RET





;============================================================================





INIT_T2:


; inicjalizacja licznika T2 dla watchdoga


; przepelnienie bedzie sie pojawiac co ok. 70ms


; jednobajtowy rejestr licznika przepelnien WD da mozliwe opoznienie do ok 18s


	CLR A


	MOV RCAP2L,A			; ustawienie wartosci poczatkowej dla T2


	MOV RCAP2H,A			; wartoscia 0000H


	MOV WD,A			; licznik pomocniczy przepelnien T2 dla WATCHDOGa


	SETB T2CON.0			; praca z automatycznym wpisywaniem 


	CLR T2CON.1			; jako licznik impulsow wewnetrznych


	CLR EXEN2			; zablokowanie wejscia T2EX


	CLR TCLK


	CLR RCLK


	SETB ET2			; odblokowanie przerwania od T2


	SETB TR2			; odblokowanie licznika


	SETB EA


	RET





;============================================================================





T2_SERVICE:


; obsluga przerwania timera T2


; tylko zapalanie flagi RS_TIMEOUT





	CLR EA


	CLR TF2


	PUSH ACC


	PUSH AR0





; if NOP_SENT==1 && RS_TIMEOUT==1 then DAEMON_ERROR


	JNB NOP_SENT,T2S00


	JNB RS_TIMEOUT,T2S00


	MOV A,#CLEAR_DISP


	LCALL WRITE_COMMAND


	MOV A,#0C0H


	LCALL WRITE_COMMAND


	MOV DPTR,#MSG3


	LCALL PRINT_STR_C		; "Daemon failure" na LCD


	MOV A,#0D0H


	LCALL WRITE_COMMAND


	MOV DPTR,#MSG2


	LCALL PRINT_STR_C		; "Press any key" na LCD





T2S100:	LCALL READKB


	JZ T2S100


	MOV A,#CLEAR_DISP


	LCALL WRITE_COMMAND


	CLR TR2				; blokujemy licznik T2 


	LJMP START





; if RS_TIMEOUT==1 then WRITE_RS(NOP)





T2S00:	JNB RS_TIMEOUT,T2S0


; wysylamy NOP na piechote


	SETB REN


	MOV SBUF,#'N'


	JNB TI,$


	CLR TI


	MOV SBUF,#00H


	JNB TI,$


	CLR TI


	SETB NOP_SENT


	SJMP T2S01			; dalej tak jaj przy RS_ACTIVE==1


	


; if RS_ACTIVE==1 


T2S0:	JNB RS_ACTIVE,T2S1





;then WD=0; RS_ACTIVE=0; RS_TIMEOUT=0


T2S01:	CLR RS_ACTIVE


	CLR RS_TIMEOUT


	MOV WD,#00H


	SJMP T2S10





;else WD=WD+1;


T2S1:


	MOV R0,#WD


	MOV A,@R0


	INC A


	MOV @R0,A


; if  WD==MAX_TIME then RS_TIMEOUT=1


 


	CJNE A,#MAX_TIME,NO_TIMEOUT


	SETB RS_TIMEOUT


	MOV @R0,#00H





	SJMP T2S10


NO_TIMEOUT:


	MOV @R0,A


	


T2S10:


	POP AR0


	POP ACC


	SETB EA


	RETI





;============================================================================





INIT_RS:


; incjalizacja rs232c dla 1200 baud





	MOV SCON,#01010000B		; ustawienie trybu 1 i inicjacja odbiornika 


	ANL PCON,#7Fh			; pcon.7=0  dla 600 - 9600 bodów, pcon.7=1 dla 19200 bodów  


	ORL TMOD,#20h			; licznik T1 pracuje w trybie 2 (8 bitowy z automatycznym ładowaniem)


	MOV TH1,#0E8h			; wartości początkowe dla poszczególnych prędkości transmisji: 


	MOV TL1,#0E8h			; D0h-600, E8h-1200, F4h-2400, FAh-4800, FDh-9600, FDh-19200 


	CLR TF1


	CLR RS_FLAG.1			; zerowanie flagi wyslania znaku


	SETB TR1


	SETB ES                		; zezwolenie na przerwania transmisji szeregowej portu RS


	SETB EA				; globalne zezwolenie na przerwania


	MOV A,#RS_BUF			; poczatek bufora odbiornika


	MOV RS_PTR,A			; wskaznik bufora na pocztek


	CLR MSG_OK			; zerowanie znacznika odebrania calego komunikatu


	SETB MSG_SENT			; znacznik wyslania ostatniego komunikatu


	RET





;============================================================================





WRITE_RS:


; procedura inicjujaca wyslanie komunikatu przez rs


; komunikat musi byc zakonczony znakiem 00h


; we:	R0 - adres poczatku komunikatu w wewnetrznej pamieci RAM


; powrot do programu nastepuje po wyslaniu calego komunikatu





	CLR REN				; blokada odbiornika (na wszelki wypadek)


	PUSH AR2


	CLR MSG_SENT			; zerujemy flage wyslania komunikatu


	MOV A,@R0			; znak do wyslania do a


	MOV SBUF,A			; wysylamy pierwszy znak komunikatu


	INC R0		 


	JNB MSG_SENT,$			; oczekiwanie na zakonczenie wysylania komunikatu


	SETB REN			; odblokowanie odbiornika


	POP AR2


	RET


 	     


;============================================================================





RS_INT:


; obsluga przerwania portu szeregowego


	CLR EA				; blokada przerwan


	PUSH PSW


	PUSH ACC





	JB TI,RS_TX			; osluga przerwania od tx


	SETB RS_ACTIVE


; if NOP_SENT==1 


	JNB NOP_SENT,RS_RX


	CLR NOP_SENT


	CLR RS_TIMEOUT





RS_RX:		


	MOV A,SBUF			; zachowanie odebranego znaku





;	MOV R7,A


;	LCALL PRINT





	PUSH AR0	


	CLR RI				; wyzerowanie flagi odbiornika


	MOV R0,RS_PTR


	MOV @R0,A			; odebrany znak do bufora


	INC R0				; zwiekszamy wskaznik bufora


	MOV RS_PTR,R0			; zapamietanie aktualnego wskaznika bufora


	CJNE A,#00H,RSRX1		; jesli w ACC #00h - koniec komunikatu


	DEC R0


	MOV @R0,#00H


	SETB MSG_OK			; ustawienie znacznika odebrania komunikatu


	MOV R0,#RS_BUF			; ustawienie bufora na poczatek


	MOV RS_PTR,R0


RSRX1:	


	POP AR0


	POP ACC


	POP PSW


	SETB EA				; odblokowanie przerwan


	RETI





RS_TX:	CLR TI


	SETB RS_ACTIVE


	JB MSG_SENT,RSTX2		; nie ma wiecej znakow do wyslania


	MOV A,@R0			; kolejny znak z bufora


	MOV SBUF,A			; wysylamy


	JNZ RSTX1			; jesli nie koniec komunikatu to koniec obslugi


	SETB MSG_SENT			; ustawienie flagi wyslania komunikatu


	SJMP RSTX2


RSTX1:	INC R0				; wskaznik bufora o 1





RSTX2:	POP ACC


	POP PSW


	SETB EA				; odblokowanie przerwan


	RETI





;=============================================================





INIT_LCD:


;incjalizacja wyswietlacza wersja dla spawacza


	MOV P0,#0FFH


	CLR LCD_E


	CLR LCD_RS			;rej rozkazow lcd


	CLR LCD_RW


	MOV P0,#3FH


	SETB LCD_E


	NOP


	CLR LCD_E


	LCALL DELAY





	CLR LCD_RS			;rej rozkazow lcd


	CLR LCD_RW


	MOV P0,#3FH


	SETB LCD_E


	NOP


	CLR LCD_E


	LCALL DELAY





	CLR LCD_RS			;rej rozkazow lcd


	CLR LCD_RW


	MOV P0,#3FH


	SETB LCD_E


	NOP


	CLR LCD_E


	LCALL DELAY





	MOV A,#0FH


	LCALL WRITE_COMMAND		;disp=on cur=on blink=on


	MOV A,#CLEAR_DISP


	LCALL WRITE_COMMAND		


	MOV A,#06H			;entry mode set i/d=1 shift=off


	LCALL WRITE_COMMAND


	RET





PRINT:


;wyswietlenie znaku z rejestru r7





	MOV P0, #0FFH


	CLR LCD_E


	PUSH ACC


	CLR LCD_RS


	SETB LCD_RW


PR1:	SETB LCD_E


	MOV A,P0			; odczyt z rejestru statusu lcd


	CLR LCD_E


	JB ACC.7,PR1			; czekanie na gotowosc





	POP ACC


	CLR LCD_RW			; teraz piszemy


	SETB LCD_RS


	MOV P0,R7


	SETB LCD_E


	NOP


	CLR LCD_E


	RET





;=============================================================================





WRITE_COMMAND:


; wyslanie rozkazu do wyswietlacza lcd


; kod rozkazu w akumulatorze








	MOV P0, #0FFH


	PUSH ACC


	CLR LCD_RS


	SETB LCD_RW


WR_CMD:	SETB LCD_E	


	MOV A,P0			; odczyt z rejestru statusu lcd


	CLR LCD_E


	JB ACC.7,WR_CMD			; czekanie na gotowosc


	POP ACC


	CLR LCD_RW			; teraz piszemy


	MOV P0,A


	SETB LCD_E


	NOP


	CLR LCD_E


	RET





;============================================================================


PRINT_RESULT:


; wyswietlenie wyniku wykonania polecenia z uwzglednieniem zanku LF


; w R1 poczatek tekstu zakonczony #00h ze znakami LF


	


	PUSH ACC


	PUSH AR0


	MOV R0,#00H			;w R0 przechowujemy licznik linii





PR_RES0:


	MOV A,@R1


	PUSH ACC


	CLR C


	SUBB A,#'"'


	JZ PR_RE01			; koniec tekstu


	POP ACC





	CJNE A,#LF,PR_RES3		; jesli nie przejscie do nastepnej linii


	CJNE R0,#0,PR_RES5		


	PUSH ACC


	MOV A,#0C0H			; z pierwszej do drugiej


	LCALL WRITE_COMMAND


	INC R0


	POP ACC


	SJMP PR_RES2





PR_RES5:


	CJNE R0,#1,PR_RES4


	PUSH ACC


	MOV A,#090H			; z drugiej do trzeciej


	LCALL WRITE_COMMAND


	POP ACC


	SJMP PR_RES2





PR_RES4:


	SJMP PR_RES1			;  za dlugi tekst , nie bedziemy juz wiecej drukowac


	


PR_RES3:				; normalny znak


	MOV R7,A


	LCALL PRINT


PR_RES2:


	INC R1


	SJMP PR_RES0


PR_RE01:


	POP ACC


PR_RES1:


	POP AR0


	POP ACC


	RET	





;============================================================================





PRINT_STR_ID:


; wyswietlenie tekstu z wewnetrznej pamieci danych


; w r1 poczatek str


; ostatni znak tekstu 00h


; wykorzystane A, R7





	PUSH ACC


PSID1:	MOV A,@R1			; znak do A


	JZ PSID2			; koniec tekstu


	MOV R7,A


	LCALL PRINT			; wyswietl znak


	INC R1				; zwiekszenie wskaznika napisu


	SJMP PSID1


PSID2:	POP ACC


	RET	





;===========================================================





PRINT_STR_C:


; wyswietlenie tekstu z pamieci programu


; w DPTR poczatek tekstu


; ostatni znak tekstu 00h


; wykorzystane A, R7





PSC1:	CLR A


	MOVC A,@A+DPTR			; znak do a


	JZ PSC2				; koniec tekstu


	MOV R7,A


	LCALL PRINT			; wyswietl znak


	INC DPTR			; zwiekszenie wskaznika napisu


	SJMP PSC1


PSC2:	RET	





;=======================================================


DELAY:


	PUSH AR0


	PUSH AR1


	PUSH AR2


	MOV R0,#01H


QWE2:	MOV R1,#0FFH


QWE1:	MOV R2,#0FFH


	DJNZ R2,$


	DJNZ R1,QW
